# $Id: PKGBUILD 174748 2013-01-06 01:52:18Z stephane $
# Maintainer: Stéphane Gaudreault <stephane@archlinux.org>
# Contributor: Aaron Griffin <aaron@archlinux.org>

pkgname=nasm
pkgver=2.10.07
pkgrel=1
pkgdesc="An 80x86 assembler designed for portability and modularity"
url="http://www.nasm.us"
license=('BSD')
arch=('i686' 'x86_64')
depends=('glibc')
source=(http://www.nasm.us/pub/nasm/releasebuilds/${pkgver}/${pkgname}-${pkgver}.tar.bz2)
sha1sums=('2eb839c25ad0aa43a0d2d48146fd8ac708d9aabf')
install=nasm.install

build() {
  cd "${srcdir}/${pkgbase}-${pkgver}"
  ./configure --prefix=/usr
  make
  make -C doc info
}

package() {
  cd "${srcdir}/${pkgbase}-${pkgver}"
  make INSTALLROOT="${pkgdir}" install install_rdf
  install -dm755 "${pkgdir}"/usr/share/info
  install -m644 doc/info/* "${pkgdir}"/usr/share/info/
  install -D -m644 LICENSE "${pkgdir}"/usr/share/licenses/nasm/LICENSE
}
